A Quick Introduction to Maple 11

Tim McLarnan

Introduction

Maple is a tool for doing nearly anything you think of as computational in mathematics, as well as many things you may be surprised to find a machine doing. Examples include:

• Anything you can do with a calculator.

• Exact computations with fractions.

• Adding, multiplying, and factoring polynomials.

• Solving equations, either exactly or approximately.

• Simplifying algebraic expressions.

• Plotting functions.

This leaflet is a quick introduction to some of the salient things *Maple* can do. The idea has been to use the most common *Maple* functions in order to give you examples of their syntax.

Five Essential Facts

(1) The *Maple* prompt is the symbol >.

(2) *Maple* commands always end with a semicolon ;

(3) *Maple* commands are always followed by ENTER. If you want to type a command with more than one line, use SHIFT-ENTER to go down a line without executing a command.

(4) *Maple* is case sensitive. **Pi** is not the same as **pi**.

(5) Save early and often. *Maple* seems to crash most often when you are trying to print. So a recipe for disaster is to do 2 hours of work without ever saving, then try to print.

Basic Commands

+, **-**, * and / do what you expect:

```
> 2+5;
```

$$7 \qquad \textbf{(1)}$$

Unlike most calculators, however, *Maple* does operations with fractions exactly:

```
> 1/2 + 1/3 + 1/4 + 1/5 + 1/6 + 1/7;
```

$$\frac{223}{140} \qquad \textbf{(2)}$$

Other arithmetic operations are
a^b, which means *a* to the power *b*,
n!, which means *n* factorial (1*2*3*4*...*n).

```
> 2^1000;
```
$$10715086071862673209484250490600018105614048117055336074437503883703510511\backslash 249361224931983788156958581275946729175531468251871452856923140435984577\backslash 574698574803934567774824230985421074605062371141877954182153046474983581\backslash 941267398767559165543946077062914571196477686542167660429831652624386837\backslash 205668069376 \quad (3)$$

```
> (355/113)^10;
```
$$\frac{31789487802830871103515625}{3394567389922223 14849} \quad (4)$$

```
> 4!;
```
$$24 \quad (5)$$

```
> 125!;
```
$$18826771768889260997437677024916008575954036487149242588759823150835315633\backslash 161359886688293288949592313364640544593005774063016191934138059781888345\backslash 75585470555243263755650071317708800000000000000000000000000000000 \quad (6)$$

Is it surprise to you that 125! ends in so many zeros?
Can you find a way to compute how many zeros are at the end of n! for any n?

Warning: Trying anything too huge (like 10000!) may lock up your computer and needlessly antagonize your professor.

ifactor(n) factors the integer n into primes:

```
> ifactor(315);
```
$$(3)^2 (5) (7) \quad (7)$$

```
> 20! - 12!;
```
$$2432902007697638400 \quad (8)$$

The most recent result (2432902007697638400) can be easily included in any subsequent calculation without having to type it. The percent character (**%**) is used to refer to the last expression computed by *Maple*. (The last in time, not necessarily the one just above on the page.) Similarly, **%%** is the expression before **%**, and **%%%** is the expression before **%%**.

```
> ifactor(%);
```
$$(2)^{10} (3)^5 (5)^2 (7) (11)^3 (41976119) \quad (9)$$

Another way to refer to a previous result is to use the label number *Maple* places by default after each output line. Here we factor the 6th number we computed by using **Insert→Label** to insert the label (6). One has to insert labels this way. One can't just type (6) and have it work. On the bright side, the labels update automagically if you insert more equations in the middle of a worksheet.

```
> ifactor((6));
```
$$(2)^{119} (3)^{59} (5)^{31} (7)^{19} (11)^{12} (13)^9 (17)^7 (19)^6 (23)^5 (29)^4 (31)^4 (37)^3 (41)^3 (43)^2 (47)^2 (53)^2 (59)^2 (61)^2 (67) (71) (73) (79) (83) (89) (97) (101) (103) (107) (109) (113) \quad (10)$$

Decimal Approximations

Maple's normal mode of computation is exact arithmetic, with no roundoff or truncation errors.

```
> (2^30/3^20)*sqrt(2);
```

$$\frac{1073741824}{3486784401}\sqrt{2} \quad (11)$$

The function **evalf()** gives a numerical approximation to this exact expression:

```
> evalf(%);
```

$$0.4355016184 \quad (12)$$

If 10 digits aren't enough, give **evalf** an optional second argument to tell it how many digits you need:

```
> evalf(%%,60);
```

$$0.435501618497697542678003469998073498141904412675140022701397 \quad (13)$$

Maple uses **Pi**, not **pi**, to represent π. Notice what this means: variable names in *Maple* are case-sensitive.

Would you like to know the first few digits of π?

```
> evalf(Pi,200);
```

3.14159265358979323846264338327950288419716939937510582097494459230781640628\
8620899862803482534211706798214808651328230664709384460955058223172535940\
4081284811174502841027019385211055596446229489549303820 **(14)**

Even more surprisingly, *Maple* can take many decimal numbers and find simple expressions that they might equal. Here's an example: Suppose I do a calculation and get the result 2.985009889. I may look at that number and not recognize it at all. *Maple,* however, does recognize it:

```
> identify(2.985009889);
```

$$\sqrt{2}+\frac{1}{2}\pi \quad (15)$$

Knowing that my number is at least close to $\sqrt{2}+\frac{\pi}{2}$ gives me a useful fact about my number, that perhaps I will then be able to prove.

Variable Expressions

Maple also works with variables:

```
> (2*x+7) * (x^2+3) * (x+2)^5;
```

$$(2x+7)(x^2+3)(x+2)^5 \quad (16)$$

```
> expand(%);
```

$$2x^8+27x^7+156x^6+521x^5+1170x^4+1944x^3+2384x^2+1872x+672 \quad (17)$$

At this point, *Maple* has forgotten where it got this polynomial. It's just a random polynomial of degree 8. But *Maple* can factor it:

```
> factor(%);
```

$$(2x+7)(x^2+3)(x+2)^5 \quad (18)$$

Could you have factored this polynomial as fast as *Maple* did? Could you have factored it at all? (If you think about it a bit, your answers to these questions should be "no" and "yes.")

Maple doesn't always get the terms of a polynomial in the order you would expect:

```
> expand((1+x)^3*(x+2)^2);
```

$$25 x^2 + 16 x + 4 + 19 x^3 + 7 x^4 + x^5 \quad (19)$$

In this case, you can use the **sort** command to get things arranged.

```
> sort(%);
```

$$x^5 + 7 x^4 + 19 x^3 + 25 x^2 + 16 x + 4 \quad (20)$$

Assignment and Simplification

Often we need to assign a name to the result of a computation. *Maple* does this using the syntax **variable := value;**

In making an assignment, *Maple* does some obvious simplifications first.

```
> e1 := (x+y)^3 * (x+y)^2;
```

$$e1 := (x + y)^5 \quad (21)$$

Maple has a built in function called **simplify**, which tries to simplify expressions. It does not always find the simplest form of an expression, but it is a start. Here is an example:

```
> e2 := (x^3-y^3)/(x^2+x-y-y^2);
```

$$e2 := \frac{x^3 - y^3}{x^2 + x - y - y^2} \quad (22)$$

```
> simplify(e2);
```

$$\frac{x^2 + y x + y^2}{x + 1 + y} \quad (23)$$

To understand what went on here, we can use the commands **numer** and **denom** (to take the numerator and denominator of a fraction) together with **factor**:

```
> factor(numer(e2));
```

$$(x - y) \left(x^2 + y x + y^2\right) \quad (24)$$

```
> factor(denom(e2));
```

$$(x + 1 + y) (x - y) \quad (25)$$

So what **simplify** did was to find and remove the common factor of $x - y$. In this particular case, the commands **factor** and **simplify** do the same thing:

```
> factor(e2);
```

$$\frac{x^2 + y x + y^2}{x + 1 + y} \quad (26)$$

Solving Equations

Maple can also solve equations, even symbolic ones. Notice the syntax of the commands below. You have to specify first the equation, then the variables you want to solve for.

```
> solve(x^2+5*x+2 = 0, x);
```

$$-\frac{5}{2} + \frac{1}{2} \sqrt{17}, -\frac{5}{2} - \frac{1}{2} \sqrt{17} \quad (27)$$

This quadratic equation had two solutions (separated by a comma), just as you would expect.

If you want a decimal approximation instead of an exact solution to an equation, use **fsolve** instead of **solve**.

```
> fsolve(x^2+5*x+2 = 0, x);
```

$$-4.561552813, -0.4384471872 \quad (28)$$

solve and **fsolve** can also deal with systems of more than one equation. You give them a set of equations (in set brackets { }) followed by a set of variables to solve for. Here's an example:

```
> solve({x+y=5, x-y=2}, {x,y});
```

$$\left\{y=\frac{3}{2}, x=\frac{7}{2}\right\} \quad (29)$$

Now let's get more adventuresome and try to solve an equation containing symbolic constants:

```
> solve(a*x^2+b*x+c=0, x);
```

$$-\frac{1}{2}\,\frac{b-\sqrt{b^2-4\,a\,c}}{a}, -\frac{1}{2}\,\frac{b+\sqrt{b^2-4\,a\,c}}{a} \quad (30)$$

Do you recognize this answer? It's the quadratic formula!
Did you know that there was a similar formula for solving cubic equations? It's a bit more complicated: its output will fill the screen.

```
> solve(a*x^3 + b*x^2 + c*x + d = 0, x);
```

$$\frac{1}{6}\,\frac{1}{a}\left(36\,c\,b\,a-108\,d\,a^2-8\,b^3 \quad (31)\right.$$

$$\left.+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\right)^{1/3}-\frac{2}{3}\left(3\,a\,c\right.$$

$$\left.-b^2\right)\Big/\left(a\left(36\,c\,b\,a-108\,d\,a^2-8\,b^3\right.\right.$$

$$\left.\left.+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\right)^{1/3}\right)-\frac{1}{3}\,\frac{b}{a},$$

$$-\frac{1}{12}\,\frac{1}{a}\left(36\,c\,b\,a-108\,d\,a^2-8\,b^3\right.$$

$$\left.+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\right)^{1/3}+\frac{1}{3}\left(3\,a\,c\right.$$

$$\left.-b^2\right)\Big/\left(a\left(36\,c\,b\,a-108\,d\,a^2-8\,b^3\right.\right.$$

$$\left.\left.+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\right)^{1/3}\right)-\frac{1}{3}\,\frac{b}{a}$$

$$+\frac{1}{2}\,\mathrm{I}\sqrt{3}\left(\frac{1}{6}\,\frac{1}{a}\left(36\,c\,b\,a-108\,d\,a^2-8\,b^3\right.\right.$$

$$\left.+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\right)^{1/3}+\frac{2}{3}\left(3\,a\,c\right.$$

$$-b^2)\Big/\Big(a\big(36\,c\,b\,a-108\,d\,a^2-8\,b^3$$

$$+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\big)^{1/3}\Big)\Big),$$

$$-\frac{1}{12}\frac{1}{a}\big(36\,c\,b\,a-108\,d\,a^2-8\,b^3$$

$$+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\big)^{1/3}+\frac{1}{3}\big(3\,a\,c$$

$$-b^2)\Big/\Big(a\big(36\,c\,b\,a-108\,d\,a^2-8\,b^3$$

$$+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\big)^{1/3}\Big)-\frac{1}{3}\frac{b}{a}$$

$$-\frac{1}{2}\,\mathrm{I}\sqrt{3}\Bigg(\frac{1}{6}\frac{1}{a}\big(36\,c\,b\,a-108\,d\,a^2-8\,b^3$$

$$+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\big)^{1/3}+\frac{2}{3}\big(3\,a\,c$$

$$-b^2)\Big/\Big(a\big(36\,c\,b\,a-108\,d\,a^2-8\,b^3$$

$$+12\sqrt{3}\sqrt{4\,a\,c^3-c^2\,b^2-18\,c\,b\,a\,d+27\,d^2\,a^2+4\,d\,b^3\,a}\big)^{1/3}\Big)\Bigg)$$

Look at this answer a bit until you undertand the notation. There are three roots, which are written separated by commas. The second and third roots also involve $\sqrt{-1}$, which *Maple* writes as **I**.

Warning again: I isn't **i**. Case matters to *Maple.*

Maple can also solve the general fourth degree equation (add in an x^4 term), though the output now fills several screens. Can it do more?

```
> solve(a*x^5 + b*x^4 + c*x^3 + d*x^2 + e*x + f = 0, x);
```

$$\mathrm{RootOf}\left(a\,\_Z^5+b\,\_Z^4+c\,\_Z^3+d\,\_Z^2+e\,\_Z+f\right) \quad \textbf{(32)}$$

Why can't *Maple* solve this equation? Because Évariste Galois proved shortly before his death at the age of 20 that there is not general formula like the quadratic formula which solves polynomial equations of degree 5 and higher. (That is, there is no formula solving all such equations. For particular values of the coefficients, there are often simple solutions.)

fsolve still works to find approximate solutions to equations of high degree, though:

```
> fsolve(x^5+x+1=0, x);
```

$$-0.7548776662 \quad \textbf{(33)}$$

Plotting Graphs

Maple only found one solution to the last equation. How can we convince ourselves there is only one? One way to begin to build evidence might be to graph the function.

```
> plot(x^5+x+1, x=-2..2);
```



This at least suggests there is only one root between x=2 and x=2.
Maple sometimes makes an unhelpful choice of axes:

```
> plot(x^3 - 1/x^2, x=-2..2);
```



The function has a vertical asymptote. *Maple's* plot shows y values down to −1,400,000. You can fix this problem by selecting your own scale for the y-axis:

```
> plot(x^3 - 1/x^2, x=-2..2, y=-10..10);
```



There are many options for the **plot** command, and many other types of plots.
(Try typing **?plot[options];** or **?plots;** to learn more about these.)
Here is an example of a plot of several functions at once.

```
> plot({x, x^2, x^3, x^4, x^5}, x=-1.1 .. 1.1);
```



One problem with this plot is that you may not be sure which curve is which. To make matters worse, *Maple* doesn't always assign the same colors to curves: the next time you run it, the colors might change. Here's a way to deal with that: make the functions an ordered list (in square brackets) instead of an unordered set (in curly braces), and give an additional argument listing the colors you want for each curve.

To plot sin(x) in black, cos(x) in red, and arctan(x) in blue, you could say

```
> plot([sin(x),cos(x),arctan(x)], x=-Pi..3*Pi, y=-2..2, color=
  [black, red, blue]);
```



If you're color blind or if you're printing in black and white, you can use linestyles instead of colors. Again, **?plot[options];** will give useful advice.

```
> plot([sin(x),cos(x),arctan(x)], x=-Pi..3*Pi, y=-2..2, color=
  black, linestyle=[SOLID, DASH, DOT]);
```



3D plots are also possible.

```
> plot3d(x^2-y^2, x=-2..2, y=-2..2);
```



(To obtain this picture, I ran the *Maple* command, clicked on the picture, and then played with the choices on the Tool Bar until I got the view I wanted.) One can also view this 3-dimensional surface as a contour plot. You can make contour plots from 3D plots using the Tool Bar.

Help!

In order to use *Maple* effectively, you'll need to know where to get help. An excellent and convenient source is *Maple's* online help, which is available under the Help menu item. You can also access help directly from the *Maple* prompt, like this:

```
> ?coeffs;
```

The output appears in another window, but here it is. It includes a calling sequence, usually some obtuse discussion. Most usefully, it ends with examples and with suggestions of other related functions to look at. All functions and libraries which *Maple* knows about have online help items. The online help is arranged hierarchically; so if you don't know the name of the command you're looking for, you can often find it by browsing through the help system

coeffs - extract all coefficients of a multivariate polynomial

Calling Sequence

coeffs(p, x, 't')

Parameters

p - multivariate polynomial
x - (optional) indeterminate or list/set of indeterminates
t - (optional) an unevaluated name

Description

- The coeffs function returns an expression sequence of all the coefficients of the polynomial p with respect to the indeterminate(s) x.
- If x is not specified, coeffs computes the coefficients with respect to all the indeterminates of p (see the indets function). If a third argument t is specified (call by name), it is assigned

an expression sequence of the terms of p. There is a one-to-one correspondence between the coefficients and the terms of p.

- Note that p must be collected (collect) with respect to the appropriate indeterminates. For multivariate polynomials, use collect with `distributed`.

Examples

```
> s := 3*v^2*y^2+2*v*y^3;
```

$$s := 3\,v^2\,y^2 + 2\,v\,y^3 \quad (34)$$

```
> coeffs( s );
```

$$3, 2 \quad (35)$$

```
> coeffs( s, v, 't' );
```

$$3\,y^2, 2\,y^3 \quad (36)$$

```
> t;
```

$$v^2, v \quad (37)$$

```
> p := a*x^2+b*x*y+c*x^2+d*x+e*y+f:
coeffs(collect(p,[x,y],'distributed'),[x,y]);
```

$$f, b, a + c, e, d \quad (38)$$

See Also

collect, coeff, tcoeff, lcoeff, indets, PolynomialTools[CoefficientVector]

The bookshelf in the Math/CS Lounge on the second floor of Dennis contains a fair number of manuals for various versions of *Maple*. The basics of the program haven't changed much, so any of these manuals should get you started. Other students, faculty, etc. are also good resources. Finally, the *Maple* Help menu includes, among other things, a couple of pretty complete manuals.

Substitutions, Etc.

Here are a few more commands which are useful in calculus. We begin by defining an expression, making some substitutions, and computing its derivative directly from the definition:

```
> f := x^2+x;
```

$$f := x^2 + x \quad (39)$$

```
> subs(x=3, f);
```

$$12 \quad (40)$$

```
> subs(x=x+h, f);
```

$$(x + h)^2 + x + h \quad (41)$$

```
> expand(%);
```

$$x^2 + 2\,x\,h + h^2 + x + h \quad (42)$$

```
> % - f;
```

$$2\,x\,h + h^2 + h \quad (43)$$

```
> %/h;
```

$$\frac{2\,x\,h + h^2 + h}{h} \quad (44)$$

```
> simplify(%);
```

$$2\,x+h+1 \tag{45}$$

```
> limit(%, h=0);
```

$$2\,x+1 \tag{46}$$

Defining Functions

An irritating feature of this calculation was using **subs(x=3, f)** to compute f when $x=3$. We'd like to be able just to say **f(3)**, but we can't, because **f** is just an expression, not a function. So how would we define a function in *Maple*? There are 2 ways. For simple one-line functions, we can do this:

```
> g := x -> x^2;
```

$$g := x \rightarrow x^2 \tag{47}$$

This defines g as the function taking x to x^2. We can now use g just like any other function:

```
> g(x);
```

$$x^2 \tag{48}$$

```
> g(3);
```

$$9 \tag{49}$$

```
> g(x+h);
```

$$(x+h)^2 \tag{50}$$

More complicated functions are defined in a more complicated way. Begin the function with **proc(*variables*)**, and end it with **end proc;**. Here, for instance, is a function which takes 2 arguments, and returns the larger of the two, assuming both are numbers:

```
> h := proc(x,y)
          if x>y
            then x
          else
            y
          end if
        end proc;
```

$$h := \mathbf{proc}(x, y)\ \ \mathbf{if}\ y < x\ \mathbf{then}\ x\ \mathbf{else}\ y\ \mathbf{end\ if}\ \mathbf{end\ proc} \tag{51}$$

```
> h(3,5);
```

$$5 \tag{52}$$

```
> h(5,3);
```

$$5 \tag{53}$$

More Calculus Functions

Maple can directly compute derivatives, sums, integrals and limits. Here are some examples:

```
> diff(sin(x^2),x);
```

$$2\cos\left(x^2\right)x \tag{54}$$

```
> int(x^2, x);
```

$$\frac{1}{3}x^3 \tag{55}$$

```
> int(x^2, x=1..4);
```
$$21 \quad (56)$$

```
> sum(k^2, k=1..10);
```
$$385 \quad (57)$$

```
> sum(1/k^2, k=1..infinity);
```
$$\frac{1}{6}\pi^2 \quad (58)$$

```
> y := limit((x^2-3*x+2)/(x-1), x=1);
```
$$y := -1 \quad (59)$$

Particularly when you are working with second and third derivatives and higher, it is often useful to define a function (rather than an expression) and then to use **D** to compute a derivative function. Here's a function and its first, second, and third derivatives, at *x* and at a numerical point:

```
> f := x -> x^5+1/x;
```
$$f := x \rightarrow x^5 + \frac{1}{x} \quad (60)$$

```
> f(x);
```
$$x^5 + \frac{1}{x} \quad (61)$$

```
> f(1);
```
$$2 \quad (62)$$

```
> D(f);
```
$$x \rightarrow 5\,x^4 - \frac{1}{x^2} \quad (63)$$

```
> D(f)(x);
```
$$5\,x^4 - \frac{1}{x^2} \quad (64)$$

```
> D(f)(1);
```
$$4 \quad (65)$$

```
> (D@@2)(f)(x);
```
$$20\,x^3 + \frac{2}{x^3} \quad (66)$$

```
> (D@@2)(f)(1);
```
$$22 \quad (67)$$

```
> (D@@3)(f)(x);
```
$$60\,x^2 - \frac{6}{x^4} \quad (68)$$

```
> (D@@3)(f)(1);
```
$$54 \quad (69)$$

Oops!

Like any powerful tool, Maple offers any number of ways for you to make mistakes. Here are some particularly popular ones

Mistake 1: Forgetting you have assigned a value to a variable.:

Right now, for instance, y has a value: it is -1. I'll therefore get into loads of trouble if I try to use y as a variable:

```
> plot(x^2, x=-1..1, y=-1..2);
Error, (in plot) unexpected option: -1 = -1 .. 2
```

To fix this, I need to tell *Maple* that y is now just the variable y again. I do that by saying **y:='y';**

```
> y;
```

$$-1 \tag{70}$$

```
> y := 'y';
```

$$y := y \tag{71}$$

```
> y;
```

$$y \tag{72}$$

Mistake 2: Forgetting a semicolon.

If you do this, *Maple* thinks the expression you want it to evaluate is not over. The current version of *Maple,* however, will go ahead and put in the semicolon after a warning:

```
> 2+6
Warning, inserted missing semicolon at end of statement
```

$$8 \tag{73}$$

Mistake 3: Order of operations.

Maple does **^** first, then ***** and **/**, then **+** and **-**. Notice the difference:

```
> x^1/2;
```

$$\frac{1}{2}x \tag{74}$$

Since *Maple* does exponentiation before division, it reads this as $x^1/2$.

```
> x^(1/2);
```

$$\sqrt{x} \tag{75}$$

This is a possible way to write $\sqrt{x}$, as is **sqrt(x)**.

If you are plotting or solving a complicated expression, it is often useful to get *Maple* to print it first, so you can be sure you have the parentheses in the right place. Then either give the expression a name or use copy and paste or use **%** to get the correct expression into your command.

```
> -b+sqrt(b^2-4*a*c)/2*a;
```

$$-b + \frac{1}{2}\sqrt{b^2 - 4\,a\,c}\;a \tag{76}$$

Nuts. That's not the root of a quadratic. Let's get the parentheses right.

```
> (-b+sqrt(b^2-4*a*c))/(2*a);
```

$$\frac{1}{2} \frac{-b+\sqrt{b^2-4\,a\,c}}{a} \quad (77)$$

That looks right. Now I can give it a name and work with it:

```
> root1 := %;
```

$$root1 := \frac{1}{2} \frac{-b+\sqrt{b^2-4\,a\,c}}{a} \quad (78)$$

```
> root2 := (-b-sqrt(b^2-4*a*c))/(2*a);
```

$$root2 := \frac{1}{2} \frac{-b-\sqrt{b^2-4\,a\,c}}{a} \quad (79)$$

```
> root1*root2;
```

$$\frac{1}{4} \frac{\left(-b+\sqrt{b^2-4\,a\,c}\right)\left(-b-\sqrt{b^2-4\,a\,c}\right)}{a^2} \quad (80)$$

```
> simplify(%);
```

$$\frac{c}{a} \quad (81)$$

Mistake 4: Leaving out * in Multiplication.

I do this all the time.

```
> x^2+2x;
Error, missing operator or `;`
> solve(x^2+bx+c=0, x);
```

$$\sqrt{-bx-c},\ -\sqrt{-bx-c} \quad (82)$$

Here we got the wrong answer because we wrote **bx**, which *Maple* reads as a new variable, rather than **b*x**, which is what we meant.

Worst of all, if you write something like **2(x)**, *Maple* interprets this as a function **2** evaluated at the point **x**, and returns just **2**. This is obviously not what you intend, and represents a stupid error in the *Maple* parser, but you need to be aware of it.

```
> 1+2(x^2);
```

$$3 \quad (83)$$

Mistake 5: Case sensitivity, or forgetting the name of a command.

Here are some examples:

```
> evalf(pi);
```

$$\pi \quad (84)$$

```
> evalf(Pi);
```

$$3.141592654 \quad (85)$$

```
> lim(sin(x)/x, x=0);
```

$$\lim\left(\frac{\sin(x)}{x}, x=0\right) \quad (86)$$

```
> limit(sin(x)/x, x=0);
```

$$1 \quad (87)$$

Mistake 6: Notation for trig functions.

This is really just a flaw in normal mathematical notation. When we write $\sin^2 x$, what we mean is *(sin(x))*^{2}. This shorthand is horribly misleading, but it is universal. Maple is only capable of understanding the second expression, though it will write meaningless strings that may look sensible if you try to use the first expression.

```
> sin^2*x;
```

$$\sin^2 x \quad (88)$$

```
> evalf(sin^2*1);
```

$$\sin^2 \quad (89)$$

```
> sin(x)^2;
```

$$\sin(x)^2 \quad (90)$$

```
> evalf(sin(1)^2);
```

$$0.7080734183 \quad (91)$$

Formatting Text

Maple can be used as a sort of mathematical word-processor, though it's a bit rough. The basic commands you need to know are:

To type text into *Maple*, hit the little **T** button near the top of the *Maple* window, or use **Insert→Text** or its keyboard shortcut. This will turn the current group into text instead of a *Maple* command.

To insert a new *Maple* command or a new block of text, go to **Insert→Execution Group**, and either insert a new block before or after the one containing the cursor. There are also keyboard shortcuts for this.

To insert superscripts or other mathematical expressions into *Maple* text, go to **Insert→2D Math** or click **Math** in the Toolbar, then play around with stuff like the Expression palette. There's lots more about this in the documentation.

You can also insert things like paragraphs, sections, and hyperlinks.

To use a variable name like a_2, use **a[2]**.

```
> a[0] + a[1]*x + a[2]*x^2;
```

$$a_0 + a_1 x + a_2 x^2 \quad (92)$$

Other User Interfaces

It's worth mentioning several other tidbits related to the *Maple* user interface.

First, if you have trouble remembering the syntax for common *Maple* commands, you can click on **View→Palette→Show All** to get a collection of palettes you can click on to insert templates for *Maple* commands in your worksheet. By default, these palettes are available to the left side of your *Maple* window.

Second, there are several other user interfaces available for *Maple*. This document is a *Maple* worksheet in which the input to each command is formatted as *Maple* input. This is the stable, old-fashioned way to use *Maple,* and it avoids having to input symbols using the palettes on the left or various keyboard shortcuts. The cost is that you have to remember the names of a bunch of commands. You can instead decide to use 2D math notation for input, either by creating a new prompt, clicking on the **Math** button in the Toolbar, and then using the palettes to find symbols, or by going to **Maple 11→Preferences→Display** to set the input display to 2D Math globally. Here are a couple of examples of how the 2D notation looks:

> $\int_0^{\pi} \sin(x)\, dx$

$$2 \qquad \textbf{(93)}$$

> $\sum_{n=1}^{\infty} \frac{1}{n^2}$

$$\frac{1}{6}\pi^2 \qquad \textbf{(94)}$$

There are also things called not *Maple* worksheets, but *Maple* documents. Documents are designed to support more extensive word-processing capabilities at the cost of using more keyboard shortcuts and of forcing input to be in 2D form, which means that some of the more advanced options of commands are not available. To make a *Maple* document, go to **File→New→Document Mode**. If you use *Maple* a lot and want to produce elegant looking documents, you may find documents a good way to go.

Finally, there is also a command-line version of *Maple*. You might want this, for instance, if you were connecting to a Linux host via ssh, and you wanted to run *Maple* without fancy graphics but without the overhead of an X-windows session. To run command-line *Maple*, type **maple** at the Linux shell prompt, or **cmaple** at the Windows command prompt. On the Mac, assuming a default install, you want to run **/Library/Frameworks/Maple.framework/Versions/11/bin/maple.** For the record, the ordinary graphical version of *Maple* is launched under Linux by typing **xmaple** at the shell prompt.

Spreadsheets, Drawings, Etc.

Want a spreadsheet in which the cells can contain any *Maple* expression? You can **Insert→Spreadsheet**. There is more information available from the Help system. The same thing is true of inserting drawings. Use **Insert→Image** to exist an existing image, or **Insert→Canvas** to insert a field on which you can draw with basic tools.

Packages

Many of *Maple's* commands live in library packages that are not loaded at startup. You can load one of the packages using the **with()** command. Loading a package lists the new functions that have been defined. Packages can also be loaded and a list of packages can be found by going to **Tools→Load Package**. Each package has an associated help file.

```
> with(plots);
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *densityplot*, *display*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *graphplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *interactiveparams*, *intersectplot*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *multiple*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *rootlocus*, *semilogplot*, *setcolors*, *setoptions*, *setoptions3d*, *spacecurve*, *sparsematrixplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*] **(95)**

```
> fieldplot([y,-x], x=-2..2, y=-2..2);
```




```
> with(Student[Calculus1]);
```

[*AntiderivativePlot*, *AntiderivativeTutor*, *ApproximateInt*, *ApproximateIntTutor*, *ArcLength*, *ArcLengthTutor*, *Asymptotes*, *Clear*, *CriticalPoints*, *CurveAnalysisTutor*, *DerivativePlot*, *DerivativeTutor*, *DiffTutor*, *ExtremePoints*, *FunctionAverage*, *FunctionAverageTutor*, *FunctionChart*, *FunctionPlot*, *GetMessage*, *GetNumProblems*, *GetProblem*, *Hint*, *InflectionPoints*, *IntTutor*, *Integrand*, *InversePlot*, *InverseTutor*, *LimitTutor*, *MeanValueTheorem*, *MeanValueTheoremTutor*, *NewtonQuotient*, *NewtonsMethod*, *NewtonsMethodTutor*, *PointInterpolation*, *RiemannSum*, *RollesTheorem*, *Roots*, *Rule*, *Show*, *ShowIncomplete*, *ShowSteps*, *Summand*, *SurfaceOfRevolution*, **(96)** *SurfaceOfRevolutionTutor, Tangent, TangentSecantTutor, TangentTutor, TaylorApproximation, TaylorApproximationTutor, Understand, Undo, VolumeOfRevolution, VolumeOfRevolutionTutor, WhatProblem*]

```
> RiemannSum(x*(x - 2)*(x - 3), x=0..5, method = upper, output
= plot);
```




```
>
```

Packages you may find yourself wanting include:

1. plots, for all kinds of fancy plots.

2. Student[Calculus1] and various other parts of the **Student** package, for tools aimed at helping students work with new concepts.

3. LinearAlgebra, for linear algebra, logically enough. There is also a **Student[LinearAlgebra]** package.

4. DEtools, for solving and visualizing differential equations.

Short Index

Is there more to learn about *Maple*? Yes! *Maple* is a very broad and powerful system for mathematical computation. It has more than 2000 built-in functions. The large majority of these have never been used by anyone at Earlham. Here is a very quick glossary of the most important functions for a calculus student. More information on all these is available through *Maple's* online help, or in the written and human sources mentioned above:

```
+, -, *, /            Add, subtract, multiply, divide.
^                     Raise a number to a power.
!                     Factorial.
->                    Used in procedure definitions.
=                     Used in inputting equations.
:=                    Used to assign a value to a variable.
<, >, <=, >=, <>      Mean <, >, , , and , resp.
%, %%, %%%            Previous expressions.
..                    An interval.
[ ]                   List. Lists are ordered.
```

{ }	Set. Sets are unordered.
abs	Computes the absolute value of a number.
coeff	Get one coefficient of a polynomial.
collect	Collect coefficients of like powers.
combine	Combine terms into a single term.
convert	Convert from one data type to another.
cos	Cosine. Arguments are in radians.
D	Differential operator.
denom	Denominator.
diff	Differentiate.
exp(1)	e, The base for natural logs.
evalf	Evaluate an expression as a decimal approximation.
exp	Exponential function. exp(x) = E^x.
expand	Expand out an expression.
factor	Factor a polynomial.
for...do...end do	Repeat commands.
fsolve	Find approximate solutions to one or more equations.
help	Get help on a function.
if...then...else...end if	Conditionals.
ifactor	Factor an integer.
int	Integrate a function.
Int	Write down the integral, but don't evaluate. Use with changevar.
IntTutor	Student[Calculus1] command to explore ways to evaluate an integral.
limit	Compute limits.
Limit	Write doen but don't evaluate a limit.
ln, log	Compute natural logs. ln(x)=log(x)log10(x).
max	Compute the maximum of 2 or more numbers.
min	Compute the minimum of 2 or more numbers.
normal	Put a fraction in normal form. A useful special case of simplify.
numer	Numerator.
op	Pick out one part of an expression.
plot	Plot a graph.
plot3d	Plot a 3D graph.
print	Pretty-print an expression.
proc	Define a procedure.
product	Product of finitely or infinitely many terms.
quo	Quotient of 2 polynomials.
rem	Remainder of 2 polynomials.
simplify	Try to simplify an expression.
sin	Sine. The arguments to trig functions should be in radians.
solve	Solve 1 or more equations exactly.
sort	Sort a list or the terms in a polynomial.
sqrt	Square root.
Student	A set of tutorial packages. Student[Calculus1] is one of these.
sum	Sum of finitely or infinitely many terms.
Sum	Sum written down, but not worked out.
Taylor	Taylor series expansion.
value	evaluate a Sum, Limit, or Int.
with	Load a library package. Example: with(plots);